
ABSTRACT   
 
To transfer data from Microsoft Access database to SAS has never been easy.  Unlike Oracle database, 
neither SAS® PROC SQL pass-through facility nor SAS/ACCESS can directly view or retrieve data from 
Access database tables.  Users have to undergo a very complicated process in order to import data from 
Access without the help of additional software. This paper proposes an easy way to extract data from 
Access without any other additional cost. It is quick and automatic. 

INTRODUCTION 
 
Microsoft Access database is widely used in a variety of application areas. To use SAS to analyze data 
stored in Access database, the first thing is to retrieve these data and derive SAS analysis datasets. 
However, to create SAS data sets from MS Access tables is not easy. As we know,  SAS can use PROC 
SQL pass-through facility or SAS/ACCESS® libname statement to directly view or extract data from Oracle 
database, but it does not have the power to do the same thing to Access database. Several methods have 
been developed to read data from Microsoft Access tables into SAS data sets.  They are listed in Table 1. 
 
                                       Table 1.  Methods to transfer data into SAS 
 
Method Additional Software Requirements 

1. CSV  None 1. Save each Access table as csv file, then  read in by 
DATA step.  
2. Input statement should match  the column ordering and 
data type in database table. 

2. DDE None 1. Type each Access table name in DDE  statement.  
2. An input statement be written to match the layout of 
columns. 
3. Invoke MS Access and open the table file to be input. 
4. MS Access cannot have a '.' as missing value for 
numerics. 

3. PROC ACCESS SAS/Access  to PC file 
formats 

1. Need to save Access file as a Microsoft Excel 5.0/95 
Workbook file, but doesn't need input statement. 
2. Column limit is 256, long value may be truncated with an 
error message. 

4. ODBC 1. SAS/Access 
interface to ODBC. 
2. Microsoft Access 
ODBC driver. 

1. Get ODBC installed and configured. 
2. Define DSN reference to database(.mdb file) 

5. Third party 
solution 

Third party softwares, 
such as DBMS/COPY, 
DBMS/Engine. 

1. Buy and get software installed and configured. 
2. Define some parameters if necessary. 

6. PROC IMPORT SAS/Access  to PC file 
formats 

Type each table name in SAS progrma before data 
transfer. 

 
 
For CSV and DDE methods, each table name in the mdb file needs to be typed into the INFILE statement in 
sas program. All the variable names, order and informats in INPUT statement should exactly match those in 
database tables. PROC ACCESS doesn't require input statements, but it can't directly deal with Access 
tables, all Excel file names should be typed into proc access statements. These above methods are time-
consuming and programs are not useable and hard to maintain. When tables, columns and formats in 



  

database are modified,  SAS programs also need to be updated correspondingly. Obviously, they are not 
practical to hand dozens of tables in one typical study.  
 
ODBC provides a standardized means of data transfer between data systems that are ODBC compliant. 
You will need to have an ODBC driver for each of the data systems that are to share data. Although the 
transfer of information is pretty smooth once the links are properly established,, to make ODBS work is a 
fairly complex process. Third party softwares provide another choice to create SAS data sets from the 
databases. These solutions range from simple to complex and from simplistic to highly flexible. They cost 
and occupy storage space in your computer. 
 
Generally speaking, the  IMPORT procedure provides an easy and convenient way in which to transfer data 
from external data sources such as Microsoft Access files to SAS. The problem is that users still need to 
type each table name in each PROC IMPORT since it can't access multiple tables in one time.  As we know, 
SAS can import Access database tables one by one through looping if it has the list of table names for a 
specified database.  The solution is how to get the table information by SAS dynamically during execution.  
 
This paper presents a way to allow SAS to find the table information from MS Access database, then use 
IMPORT procedure to transfer all or user selected tables in the database . 
 
MICROSOFT ACCESS SYSTEM OBJECTS  
 
MS Access System Objects consist of MSysACEs, MsysObjects, MsysQueries and MsysRelationships. 
They are automatically generated when new database is created. These system objects consist of several 
tables which contain important and useful information about the database. One such table is the 
MSysObjects table. The MsysObjects table contains all database objects such as tabels, queries, forms, 
reports, modules and macros.  The System Objects are hidden. To make them visible, you need perform the 
following steps.  

1. Open Microsoft Access.  
2. From the Tools menu, select the Options menu option.  
3. On the View tab, click the System Objects checkbox.  
4. Click OK.  

  
 

 
 
 
 
If you open MSysObjects table, you will find that it has a number of fields, but we will probably most often 
refer to five key fields:  
 
• Name (the name of the object)  
• Type (a number identifying the object's type) 



  

• DateCreate (the date and time the object was created)  
• DateUpdate (the last date and time the object was changed)  
• Connect And Database (information about connections and linked tables)  
 
The common values of type are given in Table 2 below. This contains the list of values that are commonly 
used by developers and are consistent over MS-Access versions 2.0, 7.0 and 8.0. 
 
  Table 2  Relationship of Objects and Type Values 
 
Objects Type Value 
Table 1 
Query 5 
Form -32768 
Report -32764 
Macro -32766 
Module -32761 
Relationships 8 
Linked Table 6 
 
Msysobjects is not accessible to SAS by default.  To grant SAS the necessary privelege to access 
MSysObjects table, we can perform the next steps. 
 

1. From the Tools menu, select the Security -> User and Group Permissions menu option  
2. Click the Permissions tab  
3. Select the Table entry in the Object Type combo box  

       4. In the Object Name listbox, select the MSysObjects entry  
       5. In the Permissions group box, check the Read Data check box  
 

 
Once MsysObjects table is set to readable, you make the system objects tables invisible again if you don't 
want them to appear on the table screen. The type values listed in Table 2 will allow us to select Access 
database tables to transfer.  
 
TABLE QUERY 
 
The following codes read Access table names from MsysObjectsand and specify the tables for data transfer. 
 



  

 
/*=================================================================* 
*  PART I. Read Access Database Table Names 
*==================================================================*/ 
 
%let dbname = s:\project\data\study12345-678.mdb ; 
%let uid    = xxxx; 
%let pwd    = *****; 
%let wgdb   = s:\project\data\cdms-workgroup-12345-678.mdw;  
 
proc import out = work.objects                                
            datatable = "msysobjects"  
            dbms = access97    replace; 
            database = "&dbname";  
 
            userid   = "&uid";  
            password = "&pwd";  
            workgpdb = "&wgdb"; 
run; 
 
data mstable (keep = name); 
  set objects; 
  where type in (1, 6) and name not like 'MSys%'; 
run; 
 
The MS Access database usually has user-level security, therefore, it is necessary to specify the following 
statements: PWD=, UID=, and WGDB=. You may remove these three statement lines if database is not 
password-protected or you'd like to input those parameters during program execution. 
 
Access database usually has two kinds of tables: created or linked. Type 6 represents linked tables. You 
may set type=1 to transfer created tables or type=6 for linked tables only. The statement of name not like 
'MSys%' is used to exclude system objects tables. 
 
A very useful feature of MSysObjects that should not be ignored is the information that the columns 
DateCreate and DateUpdate supply on your database objects. The DateCreate and DateUpdate can be 
used to select tables have been modified since a particular date or within a specific period.  
For example,  
 
if datecreate >='27OCT2003:00:00:00'dt; 
or  
if  dateupdate between '01OCT2003:00:00:00'dt and '15OCT2003:00:00:00'dt; 
 
 
 IMPORT TABLES  
 
/*=================================================================* 
*  PART II. Extract Data from Access Tables, and Make SAS Datasets 
*==================================================================*/ 
 
libname mydata 's:\project\data\sasdata'; 
 
data _null_;                                           
  set mstable end=lastrec; 
  call symput('table'||left(_n_), compress(name)); 
  if lastrec then call symput('ntable', compress(_n_)); 
run; 
 
%macro access_import; 
 
%do i=1 %to &ntable; 
  %put i=&i  table= &&table&i; 
     proc import out= mydata.&&table&i  



  

            datatable= "&&table&i"  
            dbms=access97 replace; 
            database="&dbname"; 
  
            userid="&uid";  
            password="&pwd";  
            workgpdb="&wgdb"; 
run; 
%end; 
%mend access_import; 
 
CONCLUSION 
 
This paper provided a good solution to improve the performance of IMPORT procedure to Access database. 
It shows  

• how to access MSysObjects in database and collect the table names  
• how to use the table name list to extract data automatically.  
 

Since any changes or modifications to Access database will be automatically recorded in MsysObjects, SAS 
can always get the updated information dynamically through the connection. There will be no out-of-date 
information problem without program changing when database is updated. 
 
You  can easily use the CONTENTS or DATASETS procedure to display dataset names and variable 
attributes without opening the Access database. Also, You can control and specify what tables to transfer 
before invoke the IMPORT procedure. 
 
One advantage of this approach is that it enables the SAS System to read data directly from Access tables 
and store them in a permanent place. It is more efficient to extract data and place it in a SAS data file 
instead of pulling the data when needed each time, especially when the table is large or several tables will 
be used in the same SAS programs. SAS data files are organized to provide optimal performance with 
PROC and DATA steps.  
 
Note: Use Access system table with extreme caution. Unintentional or intentional modifications of any 
information in these tables might result in an unusable database. However, Access system objects are 
unalterable by direct means. They are strictly for used by Access to manage the database. The data in them 
can be viewed and read  if they are set to unhidden, but not directly altered. 

